$Y_1 = Q$, $Y_M = P$. Niech $Z_i = Y_i \cup \{z\}$. Ciąg zbiorów $X_1, \ldots, X_N, Z_1, \ldots, Z_M$ zawiera wszystkie k-elementowe podzbiory zbioru E; każde dwa sąsiednie mają $k-1$ elementów wspólnych, ponadto $X_1 = A$, $Z_M = Z$ są zadanymi na początku zbiorami (rysunek). Kończy to dowód indukcyjny.

75. Oznaczmy: O — środek kuli, $A_1; \ldots, A_m$ — wybrane punkty sfery ($m = 1984$). O jest środkiem ciężkości układu $A_1, \ldots, A_m$, więc $\sum_i \overrightarrow{OA_i} = O$. Szukana wartość równa się

$$\frac{1}{2}\sum_{i,j} |A_iA_j|^2 = \frac{1}{2}\sum_{i,j} (\overrightarrow{A_iA_j})^2 = \frac{1}{2}\sum_{i,j} (\overrightarrow{OA_j}-\overrightarrow{OA_i})^2 = \frac{1}{2}\sum_{i,j} (R^2-2\overrightarrow{OA_i}\cdot\overrightarrow{OA_j}+R^2) =$$
$$= \sum_{i,j} (R^2-\overrightarrow{OA_i}\cdot\overrightarrow{OA_j}) = m^2R^2-\sum_i\sum_j \overrightarrow{OA_i}\cdot\overrightarrow{OA_j} = m^2R^2-\Big(\sum_i \overrightarrow{OA_i}\Big)\Big(\sum_j \overrightarrow{OA_j}\Big) = m^2R^2.$$

Mizar MSE (9)

Zbliżamy się do końca naszego kursu. Poznaliśmy już wszystkie konstrukcje językowe w Mizarze MSE. Dzisiaj pokażemy pewne techniki dowodzenia stosowane w przypadku występowania zdań alternatywnych. Zrobimy to na przykładach, które wcześniej rozwiązywaliśmy inaczej.

W przypadku, gdy chcemy dowodzić implikacji, której następnikiem jest zdanie alternatywne, możemy — i to jest wygodne — zastosować dowód nie wprost. Tak zrobiliśmy w odcinku 4. Możemy również postąpić nieco inaczej. Kiedy bowiem alternatywa jest prawdziwa? Wtedy, gdy co najmniej jeden z jej składników jest prawdziwy. Zatem jeżeli w trakcie dowodu założymy, że jeden ze składników dowodzonej alternatywy jest fałszywy i wykażemy, że drugi wtedy jest prawdziwy, to uważamy alternatywę za dowiedzioną. Weźmy przykład (por. odcinek 4)

```
FOR X,Y,Z ST NW[X,Y] HOLDS NW[Z,Y] OR NW[X,Z]
 PROOF
  LET X,Y,Z BE ULAMEK SUCH THAT A: NW[X,Y];
  ASSUME NOT NW[Z,Y];
   THEN NW[Y,Z] BY SPOJNOSC;
  HENCE NW[X,Z] BY PRZECHODNIOSC,A
 END;
```

O tym, jak korzystamy z faktu wyrażonego za pomocą alternatywy, wspominaliśmy już w odcinku 7. Jeżeli pewna teza wynika tak z jednego, jak i z drugiego członu pewnej (uzyskanej bądź założonej) alternatywy, to uważamy tę tezę za prawdziwą. Takie rozumowanie nazywamy dylematem. Alternatywa mówiąca, że każde zdanie jest albo prawdziwe, albo fałszywe jest aksjomatem klasycznego rachunku logicznego. Prawo to nazywają prawem wyłączonego środka — trzeciego wyjścia nie ma, stąd łacińskie *tertium non datur*.

Pokażemy teraz, jak dla przykładu z odcinka 4 zbudować dowód wprost, korzystając z prawa wyłączonego środka. Musimy się zgodzić, że każde dwie nazwy (np. nazwy zmiennych) albo odnoszą się do tego samego przedmiotu, albo do dwu różnych — innej możliwości nie ma.

```
FOR X,Y,Z ST NW[X,Y] & X<>Y & NW[Z,X] HOLDS NOT NW[Y,Z]
 PROOF
  LET X,Y,Z BE ULAMEK SUCH THAT
   A: NW[X,Y]  AND  B: X<>Y  AND  C: NW[Z,X];
  D: NOW ASSUME D': Z=X;
      NOT NW[Y,X] BY A,B,ANTYSYMETRIA;
     HENCE NOT NW[Y,Z] BY D'
    END;
  E: NOW ASSUME E': Z<>X;
      THEN NOT NW[X,Z] BY C,ANTYSYMETRIA;
      THEN E'': Z<>Y BY A;
      NW[Z,Y] BY A,C,PRZECHODNIOSC;
     HENCE NOT NW[Y,Z] BY E'',ANTYSYMETRIA
    END;
  THUS THESIS BY D,E
 END;
```

Powyższy dowód można znacznie skrócić. Checker Mizara MSE akceptuje również taki dowód wprost powyższego zdania:

```
FOR X,Y,Z ST NW[X,Y] & X<>Y & NW[Z,X] HOLDS NOT NW[Y,Z]
 PROOF
  LET X,Y,Z BE ULAMEK SUCH THAT
   A: NW[X,Y]  AND  B: X<>Y  AND  C: NW[Z,X];
   NOT NW[Y,X] BY A,B,ANTYSYMETRIA;
   HENCE NOT NW[Y,Z] BY C,PRZECHODNIOSC
 END;
```

Następny (tj. ostatni) odcinek poświęcimy zadaniom z zupełnie innej dziedziny niż nasze nieograniczenie, gęsto i liniowo uporządkowane ułamki. Ale to za miesiąc.

Dzisiaj jeszcze czujemy się w obowiązku podać pewną charakterystykę modułu sprawdzającego Mizara MSE (checkera). Wejście w techniczne szczegóły opisujące, jak odbywa się samo sprawdzanie w maszynie wymagałoby pewnie odrębnego kursu. Stąd podamy te wiadomości o checkerze, których można było i tak się domyślić oglądając dotychczasowe przykłady. A więc:

1. Każda tautologia rachunku zdań (jeśli jej maszynowy zapis zmieści się w komputerze) jest akceptowana przez checker bez dodatkowego uzasadnienia, a każde zdanie nie będące tautologią nie jest przez checker akceptowane. Każde zdanie rozpoczynające się od kwantyfikatora wymaga uzasadnienia.

2. Sprawdzając wynikanie wniosku z przesłanek checker uwzględnia jedynie te, które wymieniono po **by** oraz pochodzące z zahaczenia (**then**, **hence**).

3. Relacja równości (=) jest przez checker automatycznie traktowana jako relacja zwrotna, symetryczna i przechodnia.

4. Spójnik zdaniowy „równoważność" jest przez checker traktowany jako koniunkcja dwu implikacji.

5. Checker (na ogół) nie akceptuje wnioskowań, jeśli wśród przesłanek znajduje się więcej niż jedno zdanie rozpoczynające się od kwantyfikatora ogólnego.

6. Następujące dwie reguły wnioskowania klasycznego rachunku logicznego checker stosuje nieproszony:
a — prawo abstrahowania od konkretności,
b — prawo przechodzenia od ogólnego do szczególnego przypadku.

Zadania:

```
T26: FOR X,Y,Z ST X<>Y & Y<>Z & Z<>X
              HOLDS (M[X,Y,Z] OR M[X,Z,Y]) IFF (NW[X,Z] & NW[X,Y])
T27: FOR X,Y EX X',Y' ST M[X',X,Y'] & M[X',Y,Y']                    ;
T28: (EX X ST X=X) IMPLIES (EX X,Y,Z ST X<>Y & Y<>Z & X<>Z)        ;
```

dr Krzysztof PRAŻMOWSKI, dr Piotr RUDNICKI